

**GUERRE TERRESTRE :  
LES DEUX STRATÉGIES**

# SCIENCE VIE

MENSUEL N° 882

MARS 1991

- **La vie après la vie**

- **Cancer du sein : le coupable identifié**



## **LES SECRETS DU "PATRIOT"**

M 2578 - 882 - 20,00 F



145 FB - 6 FS - \$can 3.75 - 550 Ptas - 550 Esc - 28 Dh - 2.750 Dt - 4800 L - USA NYC \$ 4.25 - RCI : 1370 CFA - 940 CFP

# Une calculatrice scientifique

INFORMATIQUE AMUSANTE

**S**i les micro-ordinateurs familiaux, une fois programmés, permettent d'effectuer bien des tâches, il n'est malheureusement pas possible, la plupart du temps, de les utiliser directement en tant que simples calculatrices. Ceci est dommage car ils renferment, par construction, toutes les fonctions d'une calculatrice scientifique. Le petit programme que nous vous proposons ce mois-ci viendra donc combler ce manque. Notons que l'écran sera ici divisé

en deux dans le sens vertical. Sur sa partie gauche sera représentée la "calculette", munie de son clavier et de son afficheur, sur sa partie droite un listing des opérations effectuées s'affichera au fur et à mesure de leur frappe, de manière comparable à ce que propose une calculatrice équipée d'une petite imprimante.

Ces quelques précisions apportées, passons à l'écriture de notre programme. En premier lieu nous redéfinirons les symboles graphi-

ques nécessaires à la représentation de notre calculatrice. Deux fenêtres, la première utilisée pour "l'afficheur" de cette dernière, la seconde réservée à l'affichage du listing, seront ensuite mises en place. Cela effectué, l'écran de travail sera présenté. Après initialisation des variables, le programme pourra donc débiter. Il sera en fait constitué d'une série de sous-routines.

La première aura pour mission de faire entrer au fur et à mesure les chiffres frappés au clavier dans la mémoire de notre calculatrice. Un test permettra de distinguer un chiffre d'un symbole opératoire. En effet, dès la frappe de l'un d'entre eux, le programme quittera l'exploitation de cette sous-routine

```

10 MODE 2:CLS
20 REM *****
30 REM *
40 REM * INITIALISATION DES VARAIABLES. *
50 REM *
60 REM *****
70 LET FONC=0:LET X=0:LET Y=0
80 LET KB=0:LET OP=0:LET VIR=0
90 LET MEM=0:LET RECALL=0
100 REM *****
110 REM *
120 REM * DONNEES DES SYMBOLES TOUCHES. *
130 REM *
140 REM *****
150 DATA " M+ ", " M- ", " MR ", " MC "
160 DATA " Sq ", " x^y ", " Ln ", " lg "
170 DATA " 7 ", " 8 ", " 9 ", " + "
180 DATA " 4 ", " 5 ", " 6 ", " - "
190 DATA " 1 ", " 2 ", " 3 ", " x "
200 DATA " Ce ", " 0 ", " = ", " / "
210 REM *****
220 REM *
230 REM * REDEFINITION DES SYMBOLES GRAPHIQUES *
240 REM *
250 REM *****
260 SYMBOL AFTER 127
270 SYMBOL 128,1,3,7,7,7,3,1
280 SYMBOL 129,128,192,224,224,224,192,128
290 SYMBOL 134,255,0,0,0,0,0,0
300 SYMBOL 135,1,2,4,8,16,32,64,128
310 SYMBOL 136,128,128,128,128,128,128,128
320 SYMBOL 137,128,64,32,16,8,4,2,1
330 SYMBOL 138,0,0,0,0,0,0,0,255
340 SYMBOL 139,1,1,1,1,1,1,1,1
350 REM *****
360 REM *
370 REM * TRACAGE DE LA CALCULETTE. *
380 REM *
390 REM *****
400 FOR V=0 TO 5
410 FOR H=1 TO 4
420 READ X$
430 LOCATE 6*H,10+(2*V)
440 PEN 1:PAPER 0:PRINT CHR$(128);
450 PEN 0:PAPER 1:PRINT X$;
460 PEN 1:PAPER 0:PRINT CHR$(129);
470 NEXT H
480 NEXT V
490 FOR H=6 TO 30
500 LOCATE H,1:PRINT CHR$(134)
510 LOCATE H,21:PRINT CHR$(138)
520 NEXT H
530 FOR V=2 TO 20
540 LOCATE 5,V:PRINT CHR$(136)
550 LOCATE 30,V:PRINT CHR$(139)
560 NEXT V
570 LOCATE 5,1:PRINT CHR$(135)
580 LOCATE 30,1:PRINT CHR$(137)
590 LOCATE 30,21:PRINT CHR$(135)
600 LOCATE 5,21:PRINT CHR$(137)
610 LOCATE 9,6:PRINT "PROGRAMME"
620 LOCATE 17,8:PRINT "CALCULETTE. "
630 FOR H=8 TO 27
640 LOCATE H,2:PRINT CHR$(134)
650 LOCATE H,4:PRINT CHR$(138)
660 NEXT H
670 LOCATE 7,3:PRINT CHR$(136)
680 LOCATE 28,3:PRINT CHR$(139)
690 LOCATE 7,2:PRINT CHR$(135)
700 LOCATE 28,2:PRINT CHR$(137)
710 LOCATE 7,4:PRINT CHR$(137)
720 LOCATE 28,4:PRINT CHR$(135)
730 REM *****
740 REM *
750 REM * MISE EN PLACE DES FENETRES D'AFFICHAGE *
760 REM *
770 REM *****
780 WINDOW #2,8,27,3,3
790 WINDOW #3,50,70,1,24
800 PAPER #3,1:PEN #3,0:CLS#3
810 LOCATE #3,1,23
820 LOCATE #4,1,24
830 REM *****
840 REM *
850 REM * PRISE EN COMPTE DES TOUCHES FRAPPES. *
860 REM *
870 REM *****
880 LET K$=INKEY$:IF K$="" THEN GOTO 880
890 LET FONC=ASC(K$)
900 IF FONC=67 OR FONC=99 THEN CLS#2:PRINT #3," ---
--- Clear ----":LET X=0:LET Y=0:LET OP=0:GOTO 880
910 IF FONC=46 THEN LET VIR=1
920 IF FONC=77 OR FONC=109 THEN GOSUB 1990
930 IF FONC>47 AND FONC<58 THEN GOSUB 1020
940 IF (FONC>41 AND FONC<>46 AND FONC<48) OR (FONC>5
7 AND FONC<123) THEN GOSUB 1120
950 IF FONC=61 THEN LET Y=0:LET OP=0:PRINT #3," ---
Clear S.V. P---"
960 GOTO 880
970 REM *****
980 REM *
990 REM * CALCUL DU CHIFFRE FRAPPE AU CLAVIER. *
1000 REM *
1010 REM *****
1020 LET VALNUM=FONC-48
1030 IF VIR=0 THEN LET KB=(KB*10)+VALNUM
1040 IF VIR<>0 THEN LET KB=KB+(VALNUM/10^VIR):LET VI
R=VIR+1
1050 PRINT #2,KB
1060 RETURN
1070 REM *****
1080 REM *

```

pour mémoriser le type d'opération demandé et, éventuellement, pour l'exécuter.

Une sous-routine particulière sera associée à chaque type d'opération. A chaque fois dans le listing de notre programme nous indiquons clairement quel type d'opération elle concerne. A l'issue de l'exécution de chacune d'entre elles le résultat obtenu sera pris en compte par la variable réservée à l'affichage.

Deux dernières sous-routines se chargeront de celui-ci. La première concernera le petit afficheur de la calculatrice, la seconde l'impression du listing.

La frappe de ce programme ne doit pas poser de problème particulier. En effet, le Basic utilisé est

des plus standard. Il sera donc facile de l'adapter sur d'autres machines que l'Amstrad. Seules les instructions telles que MODE, LOCATE et SYMBOL devront être remplacées par leurs équivalentes mais nous avons déjà parlé de nombreuses fois de ce point dans ces lignes.

L'utilisation de notre calculatrice est également des plus simples. Elle s'emploie comme un appareil conventionnel. Tout chiffre frappé au clavier apparaît immédiatement sur son afficheur et la frappe d'une opération provoque sa remise à 0, sauf s'il doit afficher le résultat d'une opération, pour permettre la prise en compte du nombre suivant. Seules les touches telles que MR, MC, M+, M-, C, CE, etc. ne

sont pas accessibles sur la frappe d'une touche unique. Pour tous ces cas, il faudra en effet taper les deux lettres que comporte la touche afin que le programme soit en mesure de distinguer la fonction choisie de celles commençant par la même lettre.

Notons que notre calculatrice ne reprend pas l'ensemble des fonctions mathématiques disponibles sur l'Amstrad. Cela, tout simplement, parce que la vocation de ce programme n'est pas de refaire un ordinateur.

Il vous sera cependant parfaitement possible si vous le souhaitez, en vous inspirant des méthodes que nous avons employées, de compléter à votre gré cette calculatrice.

Henri-Pierre Penel

```

1090 REM * AGUILLAGE VERS SOUS ROUTINES OPERATION *
1100 REM *
1110 REM *****
1120 IF FONC=61 THEN GOTO 1150
1130 IF RECALL=1 THEN LET KB=MEM:LET RECALL=0
1140 IF X=0 THEN LET X=KB:LET OP=FONC:PRINT #3,X:GOT
O 1250
1150 IF Y=0 THEN LET Y=KB:PRINT #3," ";CHR$(OP):IF Y
<>0 THEN PRINT #3,Y
1160 IF OP=43 THEN GOSUB 1320
1170 IF OP=45 THEN GOSUB 1400
1180 IF OP=42 THEN GOSUB 1480
1190 IF OP=47 THEN GOSUB 1570
1200 IF OP=94 THEN GOSUB 1660
1210 IF OP=83 OR OP=115 THEN LET Y=X:GOSUB 1750
1220 IF OP=76 THEN LET Y=X:GOSUB 1830
1230 IF OP=108 THEN LET Y=X:GOSUB 1910
1240 PRINT #2,X:PRINT #3,"=";X
1250 LET KB=0:LET VIR=0
1260 RETURN
1270 REM *****
1280 REM *
1290 REM *      SOUS ROUTINE ADDITION.      *
1300 REM *
1310 REM *****
1320 LET X=X+Y
1330 LET Y=0:LET OP=FONC
1340 RETURN
1350 REM *****
1360 REM *
1370 REM *      SOUS ROUTINE SOUSTRACTION.  *
1380 REM *
1390 REM *****
1400 LET X=X-Y
1410 LET Y=0:LET OP=FONC
1420 RETURN
1430 REM *****
1440 REM *
1450 REM *      SOUS ROUTINE MULTIPLICATION. *
1460 REM *
1470 REM *****
1480 IF Y=0 THEN RETURN
1490 LET X=X*Y
1500 LET Y=0:LET OP=FONC
1510 RETURN
1520 REM *****
1530 REM *
1540 REM *      SOUS ROUTINE DIVISION.      *
1550 REM *
1560 REM *****
1570 IF Y=0 THEN RETURN
1580 LET X=X/Y
1590 LET Y=0:LET OP=FONC
1600 RETURN
1610 REM *****
1620 REM *
1630 REM *      SOUS ROUTINE PUISSANCE.      *
1640 REM *
1650 REM *****
1660 IF Y=0 THEN RETURN
1670 LET X=X^Y
1680 LET Y=0:LET OP=FONC
1690 RETURN
1700 REM *****
1710 REM *
1720 REM *      SOUS ROUTINE RACINE CARREE.  *
1730 REM *
1740 REM *****
1750 LET X=SQR(X)
1760 LET Y=0:LET OP=FONC
1770 RETURN
1780 REM *****
1790 REM *
1800 REM *      SOUS ROUTINE LOG NEPERIEN.  *
1810 REM *
1820 REM *****
1830 LET X=LOG(X)
1840 LET Y=0:LET OP=FONC
1850 RETURN
1860 REM *****
1870 REM *
1880 REM *      SOUS ROUTINE LOG DECIMAL.    *
1890 REM *
1900 REM *****
1910 LET X=LOG10(X)
1920 LET Y=0:LET OP=FONC
1930 RETURN
1940 REM *****
1950 REM *
1960 REM *      SOUS ROUTINE CONTROLE MEMOIRE. *
1970 REM *
1980 REM *****
1990 LET K$=INKEY$
2000 IF K$="+" THEN GOTO 2050
2010 IF K$="-" THEN GOTO 2060
2020 IF K$="C" OR K$="c" THEN GOTO 2070
2030 IF K$="R" OR K$="r" THEN GOTO 2080
2040 GOTO 1990
2050 LET MEM=MEM+X:GOTO 2090
2060 LET MEM=MEM-X:GOTO 2090
2070 LET MEM=0:GOTO 2090
2080 LET RECALL=1:GOTO 2090
2090 LET Y=0:LET FONC=0:PRINT #2,"      M":PRI
NT #3," ---- Clear ----"
2100 IF K$<>"R" OR K$<>"r" THEN LET X=0
2110 PRINT #3," Contenu memoire:"
2120 PRINT #3,MEM
2130 IF K$<>"R" AND K$<>"r" THEN PRINT #3," ---- C1
ear ----"
2140 IF K$="r" OR K$="R" THEN PRINT #2,MEM
2150 RETURN

```